
Symbolic and Automatic Differentiation of Languages

CONAL ELLIOTT, USA

Formal languages are usually defined in terms of set theory. Choosing type theory instead gives us lan-
guages as type-level predicates over strings. Applying a language to a string yields a type whose elements
are language membership proofs describing how a string parses in the language. The usual building blocks of
languages (including union, concatenation, and Kleene closure) have precise and compelling specifications
uncomplicated by operational strategies and are easily generalized to a few general domain-transforming and
codomain-transforming operations on predicates.

A simple characterization of languages (and indeed functions from lists to any type) captures the essential
idea behind language “differentiation” as used for recognizing languages, leading to a collection of lemmas
about type-level predicates. These lemmas are the heart of two dual parsing implementations—using (induc-
tive) regular expressions and (coinductive) tries—each containing the same code but in dual arrangements
(with representation and primitive operations trading places). The regular expression version corresponds to
symbolic differentiation, while the trie version corresponds to automatic differentiation.

The relatively easy-to-prove properties of type-level languages transfer almost effortlessly to the decidable
implementations. In particular, despite the inductive and coinductive nature of regular expressions and tries
respectively, we need neither inductive nor coinductive/bisimulation arguments to prove algebraic properties.

1 SPECIFYING LANGUAGES
Languages are usually formalized either set-theoretically as sets of strings or operationally as
parsers. Alternatively, one can use type theory, so that a language is a type-level predicate on
“strings” (lists of an arbitrary type A of “characters”). The usual language operations are defined
in Figure 1, including language union and intersection (P ∪ Q and P ∩ Q) with their identities (the
empty language ∅ and universal language 𝒰), language concatenation (P ∗ Q) and its identity (1,
containing only the empty string), single-character languages (‘ c), Kleene star (P @), and “scalar
multiplication” (s · P, which will prove useful later).

The definitions in Figure 1 reflect the logical interpretation of types under the Curry-Howard
(“propositions as types”) isomorphism, in which types represent propositions and elements (val-
ues) of a type represent proofs of the corresponding proposition [Wadler 2015]. The specific embod-
iment of Curry-Howard in Figure 1 and throughout this paper is the language Agda [Norell 2008;
Bove et al. 2009], which is both programming language and proof assistant founded on Martin-
Löf’s intuitionistic type theory [Martin-Löf and Sambin 1984]. Specifically,

78:2 Conal Elliott

Lang = A G→ Set ℓ

∅ w = ⊥

𝒰 w = ⊤

1 w = w ≡ []

‘ c w = w ≡ [ c ]

(s · P) w = s × P w

(P ∪ Q) w = P w ⊎ Q w

(P ∩ Q) w = P w × Q w

(P ∗ Q) w = ∃ 𝜆 (u , v)→ (w ≡ u ++ v) × P u × Q v

(P @) w = ∃ 𝜆 ws→ (w ≡ concat ws) × All P ws

Fig. 1. Languages as type-level predicates

• Types have type Set ℓ for some universe level ℓ . These levels avoid logical inconsistencies
but can be safely ignored in this paper (which is parametrized over ℓ).
• The types ⊥ and ⊤ represent falsity and truth respectively, with ⊥ having no elements (i.e.,

uninhabited) and ⊤ having a single element tt.
• For types (propositions) A and B, the types A ⊎ B and A × B represent disjunction and

conjunction respectively as non-dependent sum and products.
• For a function F : A→ Set ℓ , the types ∃ F and ∀ x→ F x represent existential and universal

quantification as dependent product and function types.
• The type x ≡ y represents propositional equality (rather than computable equality) as a data

type with the single constructor refl : ∀ x→ x ≡ x.
In the definition of P@, concat flattens a list of lists into a single list via a right fold. The predicate

All P holds of a list ws when P holds for every element of ws:1

concat : (A G) G→ A G

concat = foldr _++_ []

foldr : (B→ X→ X)→ X→ B G→ X
foldr h x [] = x
foldr h x (b :: bs) = h b (foldr h x bs)

data All (P : B→ Set ℓ) : B G→ Set ℓ where
[] : All P []
_::_ : ∀ {b bs}→ P b→ All P bs→ All P (b :: bs)

These three definitions are provided in the Agda standard library [Agda Team 2020].
Note that for a language P and string w, P w is the type of proofs that w ∈ P, in other words

explanations of membership, or parsings. (If the type P w is uninhabited, then w ∉ P.) For instance,
every proof of w ∈ P ∪Q contains a proof of w ∈ P or of w ∈ Q and the knowledge of which one was
chosen (with different proofs reflecting different choices). Likewise, a proof of w ∈ P ∗ Q includes
the choice of strings u and v, a proof that w ≡ u ++ v, and proofs that u ∈ P and v ∈ Q. Figure 2
contains some examples of languages (left) and corresponding membership proofs (right).

The use of type-level predicates makes for simple, direct specification, including the use of ex-
istential quantification (as in Figure 1). As described in Section 6, it is also fairly easy to prove
algebraic properties of predicates. These definitions do not give us decidable parsing, because gen-
eral type inhabitation amounts to theorem proving. As shown in the rest of this paper, however,

1The data type constructors [] and _::_ are overloaded here, referring to both lists (as indices) and All.

Proc. ACM Program. Lang., Vol. 5, No. ICFP, Article 78. Publication date: August 2021.



Symbolic and Automatic Differentiation of Languages 78:3

a∪b : Lang
a∪b = ‘ 'a' ∪ ‘ 'b'

a∗b : Lang
a∗b = ‘ 'a' ∗ ‘ 'b'

a∪b@ : Lang
a∪b@ = a∪b @

_ : a∪b [ 'b' ]
_ = inj2 refl

_ : a∗b ('a' :: 'b' :: [])
_ = ([ 'a' ] , [ 'b' ]) , refl , refl , refl

_ : a∪b@ ('a' :: 'b' :: 'a' :: [])
_ = [ 'a' ] :: [ 'b' ] :: [ 'a' ] :: []
, refl
, inj1 refl :: inj2 refl :: inj1 refl :: []

Fig. 2. Language examples2

type-level predicates serve as a simple, precise, and non-operational basis for specification and rea-
soning, with correct and computable parsing implementations following systematically as corol-
laries.

This paper makes the following specific contributions:
• A collection of lemmas are stated and proved about functions over lists, capturing the se-

mantic essence of Brzozowski’s syntactic technique of derivatives of regular expressions
(Section 2).
• A duality is revealed between regular expressions and tries (prefix trees) in the context of

language differentiation (Section 4). Each implementation contains exactly the same code
but in transposed forms. Both are corollaries of the semantic lemmas and are automatically
proved correct by type-checking.
• The theory and algorithms are specified in terms of proof isomorphism rather than the usual,

coarser relation of logical equivalence. As such, the lemmas for language concatenation and
Kleene closure are more informative than in previous work, capturing all parsings distinctly,
including closure of languages containing the empty string.
• The vocabulary of languages is generalized from predicates on lists to predicates on arbi-

trary types, and then re-specialized to the usual vocabulary (Section 6.1). This generaliza-
tion consists of two useful covariant applicative functors on functions (the usual one on
functions-from-X and another on functions-to-Set).
• Properties are stated first at the level of predicates, making them easy to understand and

prove, thanks to elimination of operational details. These properties are then transported to
the computable parsing implementations (Section 6.2) with no additional proofs needed.

2Some explanatory comments:

• Agda variable names can contain most characters other than spaces and parentheses, e.g., “a∪b@”.
• The alphabet type (A in Figure 1) used for these examples is Char, i.e., characters.
• Single-item lists are notated with square brackets.
• Variables defined but not used can be replaced by the anonymous pattern “_” (as in proofs on the right).
• The inj1 and inj2 functions are left and right injections into a sum type, corresponding to the left and right intro-

duction forms for logical disjunction.
• Pairing is right-associative. Since the second and third examples are existential propositions, their inhabiting proofs

are dependent pairs. Each comprises a division into substrings and a proof that the substrings concatenate to the
given test string (the language’s argument), along with language membership proofs for each substring.
• Equality proofs in these examples are by normalization and hence given by refl.

78:4 Conal Elliott

Although many proofs are elided below, the source code for this paper is freely available and con-
tains fully verified Agda code with no postulates [Elliott 2021]. The type signatures and definitions
displayed in the paper are extracted from this source code by the Agda compiler.

2 DECOMPOSING LANGUAGES
In order to convert languages into parsers, it will help to understand how language membership re-
lates to the algebraic structure of lists, specifically the monoid formed by _++_ and []. Generalizing
from languages to functions from lists to any type, define

𝜈 : (A G→ B)→ B -- “nullable”

𝜈 f = f []

𝒟 : (A G→ B)→ A G→ (A G→ B) -- “derivative”

𝒟 f u = 𝜆 v→ f (u ++ v)

For languages, 𝒟 P u is the set of u-suffixes from P, i.e., the strings v such that u ++ v ∈ P.
Since u ++ [] ≡ u, it follows that3,4

𝜈◦𝒟 : 𝜈 ◦𝒟 f ⊜ f

Moreover, the function 𝒟 distributes over [] and _++_:5

𝒟[] : 𝒟 f [] ≡ f

𝒟++ : 𝒟 f (u ++ v) ≡𝒟 (𝒟 f u) v

These facts suggest that 𝒟 can be computed one list element at a time via a more specialized
operation, as is indeed the case:6

𝛿 : (A G→ B)→ A→ (A G→ B)
𝛿 f a = 𝒟 f [ a ]

𝒟foldl :𝒟 f ⊜ foldl 𝛿 f

Each use of 𝛿 thus takes us one step closer to reducing general language membership to nulla-
bility. This simple observation is the semantic heart of the syntactic technique of derivatives of
regular expressions as used for efficient recognition of regular languages [Brzozowski 1964], later
extended to parsing general context-free languages [Might and Darais 2010]. Lemmas 𝜈◦𝒟 and
𝒟foldl liberate this technique from the assumption that languages are represented symbolically,
by some form of grammar (e.g., regular or context-free), as will be exploited in Section 4.3.

In terms of automata theory, derived languages are states, with P being the initial state and 𝒟 P
u the state reached from P after consuming the string u; 𝛿 is the state transition function; and 𝜈 is
the set of accepting states. Lemmas 𝜈◦𝒟 and 𝒟foldl capture the correctness of this state machine
as a recognizer for the language P.
3Recall that Agda variable names can contain most characters other than spaces and parentheses, e.g., “𝜈◦𝒟” as declared
here.
4For functions f and g, f ⊜ g is extensional equality, i.e., ∀ x→ f x ≡ g x.
5In algebraic terms, argument-flipped 𝒟 is a monoid homomorphism from lists to endofunctions.
6Repeated 𝛿 application is expressed here via a standard left fold (with X = A G→ B for 𝒟foldl):

foldl : (X→ A→ X)→ X→ A G→ X
foldl h x [] = x
foldl h x (a :: as) = foldl h (h x a) as

𝜈∅ : 𝜈 ∅ ≡ ⊥

𝜈𝒰 : 𝜈 𝒰 ≡ ⊤

𝜈∪ : 𝜈 (P ∪ Q) ≡ (𝜈 P ⊎ 𝜈 Q)

𝜈∩ : 𝜈 (P ∩ Q) ≡ (𝜈 P × 𝜈 Q)

𝜈 · : 𝜈 (s · P) ≡ (s × 𝜈 P)

𝜈1 : 𝜈 1↔ ⊤

𝜈∗ : 𝜈 (P ∗ Q)↔ (𝜈 P × 𝜈 Q)

𝜈@ : 𝜈 (P @)↔ (𝜈 P) G

𝜈 ‘ : 𝜈 (‘ c)↔⊥

𝛿∅ : 𝛿 ∅ a ≡ ∅

𝛿𝒰 : 𝛿 𝒰 a ≡𝒰

𝛿∪ : 𝛿 (P ∪ Q) a ≡ 𝛿 P a ∪ 𝛿 Q a

𝛿∩ : 𝛿 (P ∩ Q) a ≡ 𝛿 P a ∩ 𝛿 Q a

𝛿 · : 𝛿 (s · P) a ≡ s · 𝛿 P a

𝛿1 : 𝛿 1 a←→∅

𝛿∗ : 𝛿 (P ∗ Q) a←→ 𝜈 P · 𝛿 Q a ∪ 𝛿 P a ∗ Q

𝛿@ : 𝛿 (P @) a←→ (𝜈 P) G · (𝛿 P a ∗ P @)

𝛿 ‘ : 𝛿 (‘ c) a←→ (a ≡ c) · 1

Fig. 3. Properties of 𝜈 and 𝛿 for language operations

Given the definitions of 𝜈 and 𝛿 above and of the language operations in Section 1, one can
prove properties about how they relate, as shown in Figure 3. The correct-by-construction parsing
algorithms in Section 4 are corollaries of these properties.

There are three relations involved in Figure 3: propositional equality (“≡”), (type) isomorphism
(“↔”) and extensional isomorphism (“←→”). Isomorphism relates types (propositions) whose in-
habitants (proofs) are in one-to-one correspondence. Extensional (or “pointwise”) isomorphism
relates predicates isomorphic on every argument:

P←→ Q = ∀ {w}→ P w↔ Q w

The equalities in Figure 3 are all proved automatically by normalization (i.e., their proofs are
simply refl), while the other relations require a bit more work. As with all lemmas in this paper
shown with signatures only, full proofs are in the paper’s source code and are formally verified by
the Agda compiler (including during typesetting of this paper).

3 DECIDABILITY
For effective implementations, we must bridge the gap between a type (of membership proofs) and
its decidable inhabitation. Fortunately, there is a convenient and compositional way to do so. For
type A, the type Dec A contains proof of A or a proof of ¬ A (defined as usual to mean A→⊥):7

data Dec (X : Set ℓ) : Set ℓ where
yes : X→ Dec X
no : ¬ X→ Dec X

For compositionality, we will use a few operations that lift decidability of types to decidability
of constructions on those types, as shown in Figure 4. Moreover, decidability lifts naturally from
types to predicates:

7The Agda standard library contains a more efficient version of Dec [Agda Team 2020, Relation.Nullary].

78:6 Conal Elliott

⊥? : Dec ⊥
⊥? = no (𝜆 ())

_⊎?_ : Dec A→ Dec B→ Dec (A ⊎ B)
no ¬a ⊎? no ¬b = no [ ¬a , ¬b ]
yes a ⊎? no ¬b = yes (inj1 a)
_ ⊎? yes b = yes (inj2 b)

⊤? : Dec ⊤
⊤? = yes tt

_×?_ : Dec A→ Dec B→ Dec (A × B)
yes a ×? yes b = yes (a , b)
no ¬a ×? yes b = no (¬a ◦ proj1)
_ ×? no ¬b = no (¬b ◦ proj2)

_G? : Dec A→ Dec (A G)
_ G? = yes []

Fig. 4. Compositional decidability8

Decidable P = ∀ x→ Dec (P x)

With these definitions, we can succinctly formulate the problem of decidable language recogni-
tion: given a language P, construct a term of typeDecidable P. To accomplish this transformation, we
will look for decidable building blocks that mirror the predicate vocabulary defined in Section 1.

Type isomorphisms (such as those in Figure 3) play an important role in decidability, namely
that isomorphic types or predicates are equivalently decidable:

_⊳_ : B↔ A→ Dec A→ Dec B

_◀_ : Q←→ P→ Decidable P→ Decidable Q

One direction of the isomorphism proves B from A, while the other proves ¬ B from ¬ A. In fact,
logical equivalence suffices for the results in this paper, but the stronger condition of isomorphism
enables variations such as generating many parses/proofs rather than just one.

4 FROM LANGUAGES TO PARSING
4.1 Reflections
The lemmas in Figure 3 tell us how to decompose languages defined in terms of the vocabulary from
Figure 1, while the definitions in Figure 4 tell us how compute inhabitation of the resulting types,
resulting in decidable parsing. These lemmas and definitions cannot be applied automatically in
their present form, however, because languages are functions, as are 𝜈 and 𝛿 , and so are not subject
to structural inspection. An automatic solution would need some form of reflection of language-
constructing operations or of 𝜈 and 𝛿 as inspectable data.

This situation is exactly as in differential calculus, since differentiation in that setting is also
defined on functions rather than on symbolic representations. Fortunately, there are two standard
solutions for computable differentiation, commonly referred to as “symbolic” and “automatic” dif-
ferentiation. In the former, functions are represented symbolically in some suitable vocabulary,
and pattern-matching is used to apply rules of differentiation to those symbolic representations.
8A few explanatory remarks:

• 𝜆 () is the unique function from ⊥ to any type (here also ⊥), sometimes called “absurd” or “⊥-elim”.
• Recall that inj1 and inj2 are left and right injections into a sum type.
• proj1 and proj2 are left and right projections out of a product type.
• For functions f : A→ C and g : B→ C, the function [ f , g ] : A ⊎ B→ C maps inj1 a to f a and inj2 b to g b.
• _G? reflects the fact that the empty list exists for every element type (even ⊥).

data Lang : ⋄.Lang→ Set (suc ℓ) where
∅ : Lang ⋄.∅
𝒰 : Lang ⋄.𝒰
_∪_ : Lang P→ Lang Q→ Lang (P ⋄.∪ Q)
_∩_ : Lang P→ Lang Q→ Lang (P ⋄.∩ Q)
_·_ : Dec s → Lang P → Lang (s ⋄.· P)
1 : Lang ⋄.1
_∗_ : Lang P→ Lang Q→ Lang (P ⋄.∗ Q)
_@ : Lang P→ Lang (P ⋄.@)
‘ : (a : A)→ Lang (⋄.‘ a)
_◀_ : (Q←→ P)→ Lang P→ Lang Q

𝜈 : Lang P→ Dec (⋄.𝜈 P)
𝛿 : Lang P→ (a : A)→ Lang (⋄.𝛿 P a)

⟦_⟧? : Lang P→ Decidable P
⟦ p ⟧? [] = 𝜈 p
⟦ p ⟧? (a :: w) = ⟦ 𝛿 p a ⟧? w

Fig. 5. Regular expressions (inductive)

The latter solution involves re-interpreting the function-building vocabulary to construct not just
the usual functions, but also their derivatives [Griewank 1989; Griewank and Walther 2008; Elliott
2018]. This latter option is often much more efficient than the former, since it easily avoids a good
deal of work duplicated between a function and its derivative.

Sections 4.2 and 4.3 apply the symbolic and automatic strategies respectively to languages. Both
algorithms are corollaries of the lemmas in Figure 3 and are automatically proved correct by type
checking.

4.2 Symbolic Differentiation
The “symbolic differentiation” style reflects language-building vocabulary (Figure 1) into an induc-
tive data type (of modestly extended regular expressions), while 𝜈 and 𝛿 become functions defined
by pattern-matching on that data type. For convenience, use the same names as in Section 1 for
these new counterparts, and refer to the original versions via the module prefix “⋄.”. The result is
shown in Figure 5 along with a computable function ⟦_⟧? that converts a symbolic language rep-
resentation to a computable parser. Note that the _◀_ operation from Section 3 has been reified as
a new constructor alongside the more conventional language operators. The reason _◀_ must be
part of the inductive representation is the same as the other constructors, namely so that the core
lemmas (Figure 3) translate into an implementation in terms of that representation.

The meaning of any r : Lang P is defined simply as P, ignoring the representation r itself:

⟦_⟧ : Lang P→ ⋄.Lang
⟦_⟧ {P} r = P

Decidable parsing relies only on 𝜈 and 𝛿 (via ⟦_⟧?), which are defined in Figure 6, as systemat-
ically derived from the lemmas of Figure 3. These clauses are meant to be read in column-major
order, i.e., each column is a complete definition (two definitions of ten pattern-matching clauses

Proc. ACM Program. Lang., Vol. 5, No. ICFP, Article 78. Publication date: August 2021.



78:8 Conal Elliott

𝜈 ∅ = ⊥?

𝜈 𝒰 = ⊤?

𝜈 (p ∪ q) = 𝜈 p ⊎? 𝜈 q

𝜈 (p ∩ q) = 𝜈 p ×? 𝜈 q

𝜈 (s · p) = s ×? 𝜈 p

𝜈 1 = 𝜈1 ⊳ ⊤?

𝜈 (p ∗ q) = 𝜈∗ ⊳ (𝜈 p ×? 𝜈 q)

𝜈 (p @) = 𝜈@ ⊳ (𝜈 p G?)

𝜈 (‘ a) = 𝜈 ‘ ⊳ ⊥?

𝜈 (f ◀ p) = f ⊳ 𝜈 p

𝛿 ∅ a = ∅

𝛿 𝒰 a = 𝒰

𝛿 (p ∪ q) a = 𝛿 p a ∪ 𝛿 q a

𝛿 (p ∩ q) a = 𝛿 p a ∩ 𝛿 q a

𝛿 (s · p) a = s · 𝛿 p a

𝛿 1 a = 𝛿1 ◀ ∅

𝛿 (p ∗ q) a = 𝛿∗ ◀ (𝜈 p · 𝛿 q a ∪ 𝛿 p a ∗ q)

𝛿 (p @) a = 𝛿@ ◀ (𝜈 p G? · (𝛿 p a ∗ p @))

𝛿 (‘ c) a = 𝛿 ‘ ◀ ((a ?
= c) · 1)

𝛿 (f ◀ p) a = f ◀ 𝛿 p a

Fig. 6. Symbolic differentiation (column-major/patterns)

each). Correctness is guaranteed by the types of 𝜈 and 𝛿 and so is proved automatically by type-
checking Figure 6. (Recall from Section 3 that correctness is defined as constructing a term of
type Decidable P for a given language P, as satisfied by the type of ⟦_⟧?.) We could thus use any
definitions of 𝜈 and 𝛿 that type- and termination-check, although not many definitions would do
so.

In Figure 6, note the role of the isomorphisms from Figure 3. Since those isomorphisms relate 𝜈
and 𝛿 on the language-building operations to types and languages expressed in that same vocabu-
lary, we end up with a recursive algorithm.

4.3 Automatic Differentiation
Section 4.2 embodies one choice of reflecting functions into a data representation. Now consider
the dual strategy of “automatic differentiation”. This time, reflect 𝜈 and 𝛿 into a coinductive data
type of tries9, while redefining the language-building vocabulary as functions on that data type
defined by copatterns [Abel et al. 2013; Abel and Pientka 2016].10 (This program duality pattern has
been noted and investigated by Ostermann and Jabs [2018].) Again, we will use the same names

9The classic trie (“prefix tree”) data structure was introduced by Thue [1912] to represent sets of strings [Knuth 1998,
Section 6.3], later generalized to arbitrary non-nested algebraic data types [Connelly and Morris 1995], and then from sets
to functions [Hinze 2000].
10Inductive types (“data”) describe finitely large values and are processed recursively (as least fixed points) by pattern-
matching clauses that decompose arguments. Dually, coinductive types (“codata”) describe infinitely large values and are
processed corecursively (as greatest fixed points) by copattern-matching clauses that compose results. Programs on induc-
tive types are often proved by induction, while programs on coinductive types are often proved by coinduction. (As we will
see in Section 6.2, however, the simple relationship to the specification in Section 2 allows many trivial correctness proofs,
needing neither induction or coinduction.) See Gordon [1995] for a tutorial on theory and techniques of coinduction.
Tries in general represent functions, with each trie datum position corresponding to a domain value. Even when each
domain value is finitely large, there are often infinitely many of them (e.g., lists), so tries will naturally be infinite and thus
more amenable to coinductive than inductive analysis.

record Lang (P : ⋄.Lang) : Set (suc ℓ) where
coinductive
field
𝜈 : Dec (⋄.𝜈 P)
𝛿 : (a : A)→ Lang (⋄.𝛿 P a)

⟦_⟧? : Lang P→ Decidable P
⟦ p ⟧? [] = 𝜈 p
⟦ p ⟧? (a :: w) = ⟦ 𝛿 p a ⟧? w

∅ : Lang ⋄.∅
𝒰 : Lang ⋄.𝒰
_∪_ : Lang P→ Lang Q→ Lang (P ⋄.∪ Q)
_∩_ : Lang P→ Lang Q→ Lang (P ⋄.∩ Q)
_·_ : Dec s → Lang P → Lang (s ⋄.· P)
1 : Lang ⋄.1
_∗_ : Lang P→ Lang Q→ Lang (P ⋄.∗ Q)
_@ : Lang P→ Lang (P ⋄.@)
‘ : (a : A)→ Lang (⋄.‘ a)
_◀_ : (Q←→ P)→ Lang P→ Lang Q

Fig. 7. Tries (coinductive)

𝜈 ∅ = ⊥? 𝛿 ∅ a = ∅

𝜈 𝒰 = ⊤? 𝛿 𝒰 a = 𝒰

𝜈 (p ∪ q) = 𝜈 p ⊎? 𝜈 q 𝛿 (p ∪ q) a = 𝛿 p a ∪ 𝛿 q a

𝜈 (p ∩ q) = 𝜈 p ×? 𝜈 q 𝛿 (p ∩ q) a = 𝛿 p a ∩ 𝛿 q a

𝜈 (s · p) = s ×? 𝜈 p 𝛿 (s · p) a = s · 𝛿 p a

𝜈 1 = 𝜈1 ⊳ ⊤? 𝛿 1 a = 𝛿1 ◀ ∅

𝜈 (p ∗ q) = 𝜈∗ ⊳ (𝜈 p ×? 𝜈 q) 𝛿 (p ∗ q) a = 𝛿∗ ◀ (𝜈 p · 𝛿 q a ∪ 𝛿 p a ∗ q)

𝜈 (p @) = 𝜈@ ⊳ (𝜈 p G?) 𝛿 (p @) a = 𝛿@ ◀ (𝜈 p G? · (𝛿 p a ∗ p @))

𝜈 (‘ a) = 𝜈 ‘ ⊳ ⊥? 𝛿 (‘ c) a = 𝛿 ‘ ◀ ((a ?
= c) · 1)

𝜈 (f ◀ p) = f ⊳ 𝜈 p 𝛿 (f ◀ p) a = f ◀ 𝛿 p a

Fig. 8. Automatic differentiation (row-major/copatterns)

as in Section 1, referring to the original versions via the module prefix “⋄.”. The result is shown in
Figure 7.

Decidable recognition again relies only on 𝜈 and 𝛿 , which are defined for each language opera-
tion in Figure 8 (though with a problem to be noted and fixed in the next paragraph). These clauses
are meant to be read in row-major order, i.e., each row is a complete definition (ten definitions of

record Lang i (P : ⋄.Lang) : Set (suc ℓ) where
coinductive
field
𝜈 : Dec (⋄.𝜈 P)
𝛿 : ∀ {j : Size< i}→ (a : A)→ Lang j (⋄.𝛿 P a)

⟦_⟧? : Lang ∞ P→ Decidable P
⟦ p ⟧? [] = 𝜈 p
⟦ p ⟧? (a :: w) = ⟦ 𝛿 p a ⟧? w

∅ : Lang i ⋄.∅
𝒰 : Lang i ⋄.𝒰
_∪_ : Lang i P→ Lang i Q→ Lang i (P ⋄.∪ Q)
_∩_ : Lang i P→ Lang i Q→ Lang i (P ⋄.∩ Q)
_·_ : Dec s → Lang i P → Lang i (s ⋄.· P)
1 : Lang i ⋄.1
_∗_ : Lang i P→ Lang i Q→ Lang i (P ⋄.∗ Q)
_@ : Lang i P→ Lang i (P ⋄.@)
‘ : (a : A)→ Lang i (⋄.‘ a)
_◀_ : (Q←→ P)→ Lang i P→ Lang i Q

Fig. 9. Sized tries (coinductive)

two copattern-matching clauses each). Note that the clauses in Figure 8 are syntactically identical
to those in Figure 6 but are organized dually. (The compiler-generated syntax coloring differs to
reflect the changed interpretation of the definitions.)

The proof of correctness (defined by ⟦_⟧ and ⟦_⟧?, defined as before) is almost accomplished by
type- and termination-checking, but a technical problem arises. The 𝛿 (p ∗ q) clause does not satisfy
Agda’s termination checker, which cannot see that the argument 𝛿 p a in the recursive use of _∗_
is in some sense smaller than p. (Figure 8 compiles only due to suppressing termination checking
for the problematic clause with a compiler pragma.) Fortunately, this issue was already identified
and solved by Abel [2016]—also in the setting of trie-based language recognition—by using sized
types [Abel 2008; Abel and Pientka 2016]. This solution only requires giving Lang (now tries) an
index i : Size corresponding to the maximum depth to which a trie can be searched, or equivalently
the longest string that can be matched. In practice, we will work with arbitrarily deep tries, i.e.,
ones having index ∞, as in the type of ⟦_⟧?.

The modified representation is shown in Figure 9. Decidable recognition is defined exactly as
with unsized tries (Figure 8), with the sole exception of removing the compiler pragma that sup-
pressed termination checking. This time the compiler successfully proves total correctness (includ-
ing termination).

To see the parallel with automatic differentiation (AD) more clearly, note that AD implemen-
tations sample a function and its derivative together to exploit the fact that these two computa-
tions typically share much common work. This work sharing also applies when differentiating
languages. Without this optimization we have

𝒟′ : (A G→ B)→ A G→ B × (A G→ B)
𝒟′ f u = f u ,𝒟 f u

Pred A = A→ Set ℓ

pure𝑜 : Set ℓ → Pred A
pure𝑜 x a = x

map𝑜 : (Set ℓ → Set ℓ)→ (Pred A→ Pred A)
map𝑜 g P a = g (P a)

map𝑜 2 : (Set ℓ → Set ℓ → Set ℓ)→
(Pred A→ Pred A→ Pred A)

map𝑜 2 h P Q a = h (P a) (Q a)

pure𝑖 : A→ Pred A
pure𝑖 x a = a ≡ x

map𝑖 : (A→ B)→ (Pred A→ Pred B)
map𝑖 g P b = ∃ 𝜆 a→ b ≡ g a × P a

map𝑖2 : (A→ B→ C)→
(Pred A→ Pred B→ Pred C)

map𝑖2 h P Q c = ∃ 𝜆 (a , b)→ c ≡ h a b × P a × Q b

Fig. 10. Predicate operations

The potential work-sharing version exploits the close relationship between f and 𝒟 f:

�̂� : (A G→ B)→ A G→ B × (A G→ B)
�̂� f u = let f ′ = foldl 𝛿 f u in 𝜈 f ′ , f ′

Equivalence follows from a simple inductive argument, thanks to lemmas 𝜈◦𝒟 and 𝒟foldl of
Section 2:

𝒟′≡�̂� :𝒟′ f ⊜ �̂� f
𝒟′≡�̂� [] = refl
𝒟′≡�̂� (a :: as) = 𝒟′≡�̂� as

The trie representation exploits this sharing potential by weaving 𝜈 and 𝛿 into a single structure
based on common prefixes.

5 GENERALIZING FROM LANGUAGES TO PREDICATES
The language-building vocabulary defined in Section 1 can be simplified and generalized. First
note that there are two categories of operations. One category transforms the codomain (types):
∅, 𝒰, _∪_, and _∩_. The other transforms the domain (lists): 1, _∗_, _@, and ‘. The first category
applies to predicates over all types (not just over lists). Within the second category, 1, _∗_, and _@

can apply to any monoid, while ‘ is specific to lists.
Figure 10 shows two parallel collections of predicate operations, each covariantly transforming

the codomain (left) or domain (right).11 These generalized operations then specialize to language
operations as in Figure 11, which adds language complement (∁).12

The 𝜈 and 𝛿 lemmas in Figure 3 for codomain (but not domain) operations are easily generalized
as shown in Figure 12, all proved automatically by normalization. Since the decidable language
implementations in Section 4 are corollaries of 𝜈 and 𝛿 lemmas, those implementations adapt
easily to the generalized codomain operations (pure𝑜 , map𝑜 , and map𝑜 2), resulting in a somewhat
smaller implementation and broader coverage.

11The types of operations on the left can be further relaxed from Set ℓ to any codomain type, while keeping A fixed,
resulting in the usual covariant applicative functor of functions from a fixed type [McBride and Paterson 2008]. Dually, the
operations on the right (in their current generality) form a second covariant applicative functor of functions to types.
12The list-specific operations _++_, [], and concat used in Figure 1 have been generalized to monoid operations _•_ and 𝜀 .
The definitions of 1, _∗_, and _@ do not require the monoid properties, but their properties will (Section 6.1).

Lang = Pred (A G)

∅ = pure𝑜 ⊥

𝒰 = pure𝑜 ⊤

_∪_ = map𝑜 2 _⊎_

_∩_ = map𝑜 2 _×_

_·_ s = map𝑜 (s ×_)

∁ = map𝑜 ¬_

1 = pure𝑖 𝜀

_∗_ = map𝑖2 _•_

P @ = map𝑖 (foldr _•_ 𝜀) (All P)

‘ c = pure𝑖 [ c ]

Fig. 11. Languages via predicate operations

𝜈pure𝑜 : 𝜈 (pure𝑜 x) ≡ x

𝜈map𝑜 : 𝜈 (map𝑜 g P) ≡ g (𝜈 P)

𝜈map𝑜 2 : 𝜈 (map𝑜 2 h P Q) ≡ h (𝜈 P) (𝜈 Q)

𝛿pure𝑜 : 𝛿 (pure𝑜 x) a ≡ pure𝑜 x

𝛿map𝑜 : 𝛿 (map𝑜 g P) a ≡ map𝑜 g (𝛿 P a)

𝛿map𝑜 2 : 𝛿 (map𝑜 2 h P Q) a ≡ map𝑜 2 h (𝛿 P a) (𝛿 Q a)

Fig. 12. Properties of 𝜈 and 𝛿 for predicate codomain operations

6 PROPERTIES
6.1 Predicate Algebra

The basic building blocks of type-level predicates—and languages in particular—form the vocab-
ulary of a closed semiring in two different ways. The semiring abstraction has three aspects: (a) a
commutative monoid providing “zero” and “addition”, (b) a (possibly non-commutative) monoid
providing “one” and “multiplication”, and (c) the relationship between them, namely that multipli-
cation distributes over addition and zero.13 In the first predicate semiring, which is commutative
(i.e., multiplication commutes), zero and addition are ∅ and _∪_, while one and multiplication are
𝒰 and _∩_. Closure adds a star/closure operation _> with star law: x > ≈ 1 + x* x

>.
Conveniently, booleans and types form commutative semirings with all necessary proofs already

in Agda’s standard library.14 They are both closed as well. For booleans, closure maps both false

13Distribution of multiplication over zero is also known as “annihilation”.
14The equivalence relation used for types is isomorphism rather than equality. The boolean semiring is idempotent, while
the type semiring is not (in order to distinguish multiple parsings in ambiguous languages). The latter non-idempotence

and true to true, with the star law holding definitionally. For types, the closure of A is A G (the
usual inductive list type) with a simple, non-inductive proof of star.

This first closed semiring for predicates follows from a much more general pattern. Given any
two types A and B, if B is a monoid then A→ B is as well. The monoid operation _•_ lifts to the
function-level binary operation 𝜆 (f g : A→ B)→ 𝜆 a→ f a • g a. The monoid identity 𝜀 : B lifts
to the identity 𝜆 a → 𝜀. All of the laws transfer from B to A → B. Likewise for other algebraic
structures. (As always, compiler-verified proofs are in this paper’s source code.)

Looking more closely, additional algebraic structure emerges on predicates: (full) semimodules
generalize vector spaces by relaxing the associated types of “scalars” from fields to commuta-
tive semirings, i.e., dropping the requirements of multiplicative and additive inverses. Left and
right semimodules further generalize scalars to arbitrary semirings, dropping the commutativity
requirement for scalar multiplication. Again, nothing is needed from the codomain type B beyond
the properties of a semiring, so again predicates get these algebraic structures for free (already
paid for by types). Every P : Pred A is a vector in the semimodule Pred A, including the special
case of “languages”, for which A is a list type. The dimension of the semimodule is the cardinality
of the domain type A (typically infinite), with “basis vectors” having the form pure𝑖 a for a : A.
A general vector (predicate) is a linear combination (often infinite) of these basis vectors, with
addition being union and scaling defined by pairing membership proofs (as in Figure 11).

There is still more algebraic structure to be found and exploited. When our domain type A is
a monoid (as with languages, infinite streams and grids, and functions of continuous space and
time), predicates over A form a second semiring, known as “the monoid semiring” [Golan 2005],
in which zero and addition are as in the first predicate commutative semiring and semimodule,
while one and multiplication are 1 and _∗_, defined in Figures 1 and 11. In this setting, language
concatenation is subsumed by a very general form of convolution [Golan 2005; Dongol et al. 2016;
Elliott 2019]. The semimodule structure of predicates provides the additive aspect and is built en-
tirely from the codomain-transforming predicate operations defined in Figure 1 and redefined in
Figure 11. The multiplicative aspect (convolution) comes entirely from the domain-transforming
operations applied to any domain monoid. The two needed distributive properties hold for any
domain-transforming operation, whether associative or not. The formal statements and proofs of
these properties are included in this paper’s Agda source code [Elliott 2021].

Not only do types form a closed semiring, the only properties needed from types in the monoid
semiring come from the laws of closed semirings. One can thus apply the ideas in this paper to
many other useful semirings, including natural numbers (e.g., number of parses), probability dis-
tributions, and the tropical semirings (max/plus and min/plus) for optimization. When the domain
monoid commutes (multiplicatively)—e.g., for functions of space and time—the monoid semiring
does as well. Applications include power series (univariate and multivariate) and image processing
[Elliott 2019]. Research on semirings in parsing began with Chomsky and Schützenberger [1959]
and was further explored by Goodman [1998, 1999] and by Liu [2004].

6.2 Transporting Properties from Specification to Implementations
We have seen that type-level languages have useful and illuminating algebraic properties and
that the decidable implementations in Section 4 are tightly connected to languages. Must we
prove these properties again for each implementation—where there are more operational details
to manage—or do the algebraic properties somehow transfer to those implementations? This ques-
tion immediately raises a technical obstacle. Algebraic abstractions and their properties are simply

accounts for the difference between the 𝜈 and 𝛿 lemmas for P ∗ Q and P @ in Figure 3 and the corresponding rules in
previous work, as mentioned in Section 7.

inj0 : ∀ {s : A} (s′ : F s)→ ∃ F
inj0 {s} s′ = s , s′

inj1 : ∀ {g : A→ A} (g′ : ∀ {a}→ F a→ F (g a))→ (∃ F→ ∃ F)
inj1 {g} g′ (a , x) = g a , g′ x

inj2 : ∀ {h : A→ A→ A} (h′ : ∀ {a b}→ F a→ F b→ F (h a b))→ (∃ F→ ∃ F→ ∃ F)
inj2 {h} h′ (a , x) (b , y) = h a b , h′ x y

Fig. 13. Existential wrappers

typed. For example, each instance of the Monoid abstraction involves a single type 𝜏 with a binary
operation _•_ : 𝜏 → 𝜏 → 𝜏 and identity value 𝜀 : 𝜏 . This recipe accommodates the language oper-
ations in Figures 1 and 11 and their generalizations in Figure 10, but not the dependently typed,
indexed versions of languages and their operations in Figures 5 and 9, nor decidable types and
their vocabulary in Figure 4.

Fortunately, there is a simple way to encapsulate indexed types into non-indexed types, namely
existential quantification. For instance, ∃ Dec and ∃ (Lang ∞) are non-indexed types. For a type
family F :A→ Set ℓ , values of type ∃ F are pairs (a , b) with a :A and b : F a. We can also encapsulate
dependently typed operations on indexed types into simply typed operations on the (simply typed)
existentially wrapped versions of those types. For instance, consider language union from Figure 5:

_∪_ : Lang P→ Lang Q→ Lang (P ⋄.∪ Q)

Referring to the type and operations of Figures 1 and 5 (for the inductive language representation)
via the module prefixes “⋄.” and “□.” respectively, define

Lang = ∃ □.Lang

Then wrap union as follows:

_∪_ : Lang→ Lang→ Lang
(P , p) ∪ (Q , q) = P ⋄.∪ Q , p □.∪ q

Likewise for the coinductive language representation in Figure 9. The type-level language compo-
nents of the arguments and result are given explicitly here but could instead be inferred by the
compiler. Indexed operations of all arities can be systematically wrapped in this style, as shown
in Figure 13, again with automatically inferable components given explicitly for clarity. Wrapped
union can then be defined simply as _∪_ = inj2 □._∪_ and likewise for the other operations.

In addition to transporting types and operations as just described, we also need to transport prop-
erties. A crucial question is the choice of equivalence relation for the new properties. If we choose
equivalence on these dependent pairs to be semantic—i.e., equivalence of first projections (type
indices)—then all algebraic properties of those indices will hold for the existentially wrapped ver-
sions as well. This choice may at first seem like cheating, since the entire operational aspect of the
representation is ignored. The correctness of that aspect, however, is guaranteed by its dependent
typing, which anchors its meaning (no matter how cleverly implemented) to the non-operational
type index. For the language implementations of Section 4, this anchoring is guaranteed by the
types of ⟦_⟧ and ⟦_⟧? in Figures 5 and 9. (Similarly, for the decidable type constructors in Figure 4,
correctness is guaranteed by the types of the yes and no constructors in the definition of  Dec in
Section 3.) Given this choice of equivalence, properties of index types easily lift to properties of

prop1 : (∀ a→ P a)→∀ ((a , _) : ∃ F)→ P a
prop1 P (a , _) = P a

prop2 : (∀ a b→ Q a b)→∀ ((a , _) (b , _) : ∃ F)→ Q a b
prop2 Q (a , _) (b , _) = Q a b

prop3 : (∀ a b c→ R a b c)→∀ ((a , _) (b , _) (c , _) : ∃ F)→ R a b c
prop3 R (a , _) (b , _) (c , _) = R a b c

Fig. 14. Liftings of index properties to existential types

decCCS = mkClosedCommutativeSemiring ×-⊎-closedCommutativeSemiring
Dec _⊎?_ _×?_ ⊥? ⊤? _G?

symbolicCS1 = mkCommutativeSemiring (∩-∪-commutativeSemiring _) Lang _∪_ _∩_ ∅𝒰

where open Symbolic

symbolicCS2 = mkClosedSemiring ∗-∪-closedSemiring Lang _∪_ _∗_ ∅ 1 _@

where open Symbolic

automaticCS1 = mkCommutativeSemiring (∩-∪-commutativeSemiring _) (Lang ∞) _∪_ _∩_ ∅𝒰

where open Automatic

automaticCS2 = mkClosedSemiring ∗-∪-closedSemiring (Lang ∞) _∪_ _∗_ ∅ 1 _@

where open Automatic

Fig. 15. Examples of algebraic instances transported from indices to existential wrappings

existential types, as shown in Figure 14. (These definitions are partially inferable as well, e.g., from
prop3 ′ R _ _ _ = R _ _ _.) These definitions ease specification of algebraic instances for existentially
wrapped types from corresponding instances for their indices. As examples, we can lift the com-
mutative semiring of types to wrappings of the decidable counterparts from Figure 4 in Section 3,
and lift both language semirings to both decidable implementations in Section 4, as in Figure 15.

7 RELATEDWORK
The shift from languages as sets of strings to type-level predicates (i.e., proof relevance or “parsing”)
is akin to weighted automata [Schützenberger 1961; Droste and Kuske 2019] and more generally
to semiring-based parsing [Chomsky and Schützenberger 1959; Goodman 1998, 1999; Liu 2004],
noting that types (“Set” in Agda) form a (commutative) semiring (up to isomorphism). In particular,
Lombardy and Sakarovitch [2005] investigated Brzozowski-style derivatives in this more general
setting, formalizing and generalizing the work of Antimirov [1996], who decomposed derivatives
into simpler components (“partial derivatives”) that sum to the full derivative and lead to efficient
recognizers in the form of nondeterministic automata.

Most work on formal languages in functional programming has been in parsing, particularly
via parsing combinators [Hutton and Meijer 1996; Leijen and Meijer 2001; Swierstra 2008]. Such
work typically concentrates on usability, composability, and sometimes efficiency, rather than on
simple semantic specification and verifiable correctness.

The main contributions of this paper are to formalization and mechanization of language recog-
nition and parsing. There has been some work using type theory to capture languages as type-level
predicates much as in Section 1:
• Agular and Mannaa [2009] also defined a non-indexed algebraic type of regular expressions.

They also defined indexed types for 𝜈 and 𝛿 on regular expressions rather than defining the
meanings of those operations in terms of languages and then proving properties of them.
They proved a consistency property about 𝛿 , but apparently not about 𝜈 , and they note their
method’s “inability to prove the non-membership of some string in a given language”.
• Doczkal et al. [2013] formalized regular languages constructively in Coq, also in terms of

a type-level membership predicate. They then proved several classic results about finite au-
tomata, including minimization and relationships to Nerode and Myhill partitions.
• Firsov and Uustalu [2013] formalized regular expressions in Agda with a corresponding in-

ductive type of language membership proofs and related regular expressions to nondetermin-
istic finite automata (NFAs) represented as boolean matrices. They also defined operations
on that representation corresponding to those on regular expressions and proved soundness
and completeness of the NFA representation with respect to regular language membership.
• Korkut et al. [2016] defined a non-indexed algebraic data type of regular expressions together

with an indexed inductive type of proofs of membership of strings in the language denoted
by the regular expressions and a corresponding function that matches an implicitly con-
catenated stack of regular expressions. Special attention was paid to ensuring and proving
termination, particularly in regard to languages that contain the empty string. Addressing
both involved defunctionalization and translating between general regular expressions and
a restricted “standard” form.
• Traytel [2017] defined formal languages as tries, with operations via primitive corecursion

and reasoning via coinduction, all in the Isabelle/HOL proof assistant. There appears to have
been no specification higher level (less operational) than tries, leading to some rather com-
plex definitions (which, as definitions, cannot be proved correct). Language concatenation
was particularly delicate. For the same reason, proofs of even basic properties involved coin-
duction. The authors pointed out, however, that the coinductive trie formulation enabled
better proof automation than with sets of strings.
• Abel [2016] also formulated languages via tries and in particular noted the termination is-

sue mentioned in Section 4.3 and provided the sized-types solution adopted in this paper.
Relationships to automata theory and coalgebraic notions were also explored in depth. Trie
look-up yielded boolean values rather than propositions, and there was no simpler specifi-
cation of languages with respect to which the trie implementation could be specified and
proved consistent, so every operation was defined coinductively. The concatenation and clo-
sure (star) cases were especially delicate. Equality on tries was defined by strong bisimilarity,
and thus properties required corresponding machinery to state and prove.
• Baanen and Swierstra [2020] also explored correctness of regular expression matching in

Agda, but from the perspective of effects and their predicate transformer semantics. The au-
thors shared a common starting point with Korkut et al. [2016] and addressed differentiation
on regular expressions, while separating termination from partial correctness. Their regular
expression matching function targets a free monad (later to be interpreted with suitable
effect semantics) rather than propositions.

Of the previous work involving language derivatives, none appear to have based their formal
specification and proofs on the essential, non-inductive, nearly-trivial definition in terms of func-
tions of lists from Section 2. Abel [2016] mentioned this definition informally as motivation for

the more complex, coinductively defined operations on tries. Brzozowski [1964] also made clear
mention of the underlying meaning on sets of strings in his original work on regular expression
derivatives. Moreover, none of these previous investigations appear to have addressed proof iso-
morphism (distinguishing multiple parsings/explanations), and correspondingly all use transfor-
mations based on the less precise laws originally discovered by Brzozowski [1964]. Those laws
were based on the assumption that union (“addition” in both language semirings of Section 6.1)
is idempotent, which is not true for type-level language predicates related by proof isomorphism
(rather than mere logical equivalence). The lack of idempotence is crucial for going beyond rec-
ognizers to parsers (which are essentially proof-relevant recognizers), in which we might want to
extract more than one proof (parsing). The difference is mainly visible in the 𝜈 and 𝛿 lemmas for
P ∗ Q and P @ in Figure 3. As mentioned at the end of Section 3, one can simplify the development
above somewhat by replacing proof isomorphism with the coarser notion of logical equivalence.
Doing so restores union idempotence (modulo equivalence), losing proof relevance and thus re-
placing parsing by recognition. One might then also prove that the set of derivatives of a regular
language is finite (again, modulo equivalence).

